Pensieve Header: Generating the plots accompanying the QualitativeAnalysis handout.

Initialization

```
SetDirectory[
  "C:\\drorbn\\AcademicPensieve\\Classes\\12-267\\QualitativeAnalysis"];

TagProperties[_] := {};
Options[CellExport] = {
    PageWidth → 6, CellFilter → Identity,
    ExportFormat → ".pdf", ExportOptions → {}, Split → False
  };
CellExport[tag_String] := CellExport[tag, TagProperties[tag]];
CellExport[tag_String, OptionsPattern[]] := Module[
    {cells, cell},
    cells = OptionValue[CellFilter][Cases[
       NotebookGet[EvaluationNotebook[]],
       c_Cell /; FreeQ[List @@ c, Cell] && ! FreeQ[c, CellTags → tag],
       Infinity
      ]];
    If[! OptionValue[Split],
     If[Length[cells] ≥ 1,
      If[Length[cells] == 1,
       cells = Append[First[cells], PageWidth → 72 OptionValue[PageWidth]],
       cells = Cell[CellGroup[cells], PageWidth → 72 OptionValue[PageWidth]]
      ];
      Show[
       Import[Export[tag <> OptionValue[ExportFormat],
         cells, Sequence @@ OptionValue[ExportOptions]]],
       ImageSize → 120
      ]
     ],
     k = 0;
     GraphicsRow[Table[
       ++k;
       Show[
        Import[Export[
          tag <> "-" <> ToString[k] <> OptionValue[ExportFormat],
          Append[cell, PageWidth → 72 OptionValue[PageWidth]],
          Sequence @@ OptionValue[ExportOptions]
         ]],
        ImageSize → 120
       ],
       {cell, cells}
      ]]
    ]
  ];
```

J0

J0

```
{J0} = NDSolve[
    x^2 y''[x] + x y'[x] + x^2 y[x] == 0
      && y[1] == 1 && y'[1] == 0,
    y[x], {x, 1, 50}
   ];
Plot[Evaluate[y[x] /. J0], {x, 1, 50}]
```

J0




```
CellExport["J0"]
```

{J0} = NDSolve[
x^2 y''[x] + x y'[x] + x^2 y[x] == 0
&& y[1] == 1 && y'[1] == 0,
y[x], {x, 1, 50}
];
Plot[Evaluate[y[x] /. J0], {x, 1, 50}]



J0V0

```
{V0} = NDSolve[
    V''[x] + (1 + 1/(4 x^2)) V[x] == 0
     && V[1] == 1 && V'[1] == 1 / 2,
    V[x], {x, 1, 50}
   ];
Plot[Evaluate[{y[x] /. J0, V[x] /. V0}], {x, 1, 50}]
```

J0V0




```
CellExport["J0V0"]
```



```
zs = Table[BesselJZero[0, k], {k, 10}] // N
```

```
{2.40483, 5.52008, 8.65373, 11.7915,
 14.9309, 18.0711, 21.2116, 24.3525, 27.4935, 30.6346}
```

```
zs - RotateRight[zs]
```

```
{-28.2298, 3.11525, 3.13365, 3.13781,
 3.13938, 3.14015, 3.14057, 3.14083, 3.14101, 3.14113}
```

```
FindRoot[y[x] /. J0, {x, 11.79}]
```

```
{x → 12.3041}
```

BesselZeros

```
zs = x /. Table[FindRoot[y[x] /. J0, {x, λ}], {λ, 2.8, 50, 3.14}]
```

BesselZeros

```
{2.91009, 6.03123, 9.16593, 12.3041, 15.4436, 18.5839, 21.7245, 24.8654,
 28.0064, 31.1475, 34.2888, 37.43, 40.5714, 43.7127, 46.8541, 49.9956}
```

BesselZeros

```
Table[zs[[j + 1]] - zs[[j]], {j, 1, 15}]
```

BesselZeros

```
{3.12114, 3.1347, 3.13816, 3.13954, 3.14023, 3.14062, 3.14087,
 3.14103, 3.14114, 3.14123, 3.14129, 3.14133, 3.14137, 3.1414, 3.14143}
```

```
TagProperties["BesselZeros"] = {PageWidth → 8};
CellExport["BesselZeros"]
```

zs = x /. Table[FindRoot[y[x] /. J0, {x, λ}], {λ, 2.8, 50, 3.14}]
{2.91009, 6.03123, 9.16593, 12.3041, 15.4436, 18.5839, 21.7245, 24.8654,
28.0064, 31.1475, 34.2888, 37.43, 40.5714, 43.7127, 46.8541, 49.9956}
Table[zs[[j + 1]] - zs[[j]], {j, 1, 15}]
{3.12114, 3.1347, 3.13816, 3.13954, 3.14023, 3.14062, 3.14087,
3.14103, 3.14114, 3.14123, 3.14129, 3.14133, 3.14137, 3.1414, 3.14143}

Bessel

Bessel

```
GraphicsGrid[Partition[Table[
  Plot[{BesselJ[α, x], BesselY[α, x]},
   {x, 0, 1}, AxesLabel → Automatic, PlotPoints → 100,
   PlotLabel → StringReplace["y'' + 1/x y' + (1-α^2/x^2)y == 0", "α" → ToString[α]]],
  {α, 0., 1.5, 0.5}
 ], 2]]
```

Bessel




```
CellExport["Bessel"]
```



Oscillatory RSP

```
Solve[r (r - 1) + 13/2 == 0, r]
```

$$\left\{\left\{r \to \frac{1}{2} - \frac{5\,i}{2}\right\}, \left\{r \to \frac{1}{2} + \frac{5\,i}{2}\right\}\right\}$$

OscillatoryRSP

```
Sol = NDSolve[
    y''[x] - 3 y'[x] + (13/(2 x^2) + Cos[x]) y[x] == 0 && y[1] == 1 && y'[1] == 0,
    y[x], {x, ϵ = 10^-9, 1}
   ];
Plot[Evaluate[y[x] /. Sol], {x, ϵ, 1}, PlotPoints → 1000]
```

OscillatoryRSP




```
CellExport["OscillatoryRSP"]
```



PushoutForce

```
Plot[Evaluate[Table[
   y[x] /.
    DSolve[y''[x] - y[x] == 0
       && y[0] == 1 && y'[0] == c,
     y[x], x],
   {c, {1, 0, -0.9, -1, -2}}
  ]], {x, 0, 2}, AspectRatio → 2]
```

PushoutForce



```
TagProperties["PushoutForce"] = {Split → True};
CellExport["PushoutForce"]
```

```
Plot[Evaluate[Table[
   y[x] /.
    DSolve[y''[x] - y[x] == 0
      && y[0] == 1 && y'[0] == c,
     y[x], x],
   {c, {1, 0, -0.9, -1, -2}}
  ]], {x, 0, 2}, AspectRatio → 2]
```



Change of the Independent Variable

Indep1

```
a = 50; b = 40;
ψ = NDSolve[
    y''[x] - 2/x y'[x] + y[x] == 0
      && y[1] == 1 && y'[1] == 0,
    y[x], {x, 1, a}
   ];
Plot[Evaluate[y[x] /. ψ],
 {x, 1, a}, PlotRange → {-b, b}]
```

Indep1



Indep2

```
ψ = NDSolve[
    y''[z] + 1/(3 z)^(4/3) y[z] == 0
      && y[1 / 3] == 1 && y'[1 / 3] == 0,
    y[z], {z, 1, a^3 / 3}
   ];
Plot[Evaluate[y[z] /. ψ],
 {z, 1, a^3 / 3}, PlotRange → {-b, b}]
```

Indep2




```
CellExport /@ {"Indep1", "Indep2"}
```



Airy

Airy

```
Ai1 = NDSolve[y''[x] + x y[x] == 0 && y[0] == 1 && y'[0] == 0, y[x], {x, -3, 10}];
Ai2 = NDSolve[y''[x] + x y[x] == 0 && y[0] == 0 && y'[0] == 1, y[x], {x, -3, 10}];
Ai = Join[Ai1, Ai2]
```

Airy

```
{{y[x] → InterpolatingFunction[{{-3., 10.}}, <>][x]},
 {y[x] → InterpolatingFunction[{{-3., 10.}}, <>][x]}}
```

Airy

```
Plot[Evaluate[y[x] /. Ai], {x, -3, 10}]
```

Airy




```
CellExport["Airy"]
```

```
Ai1 = NDSolve[y''[x] + x y[x] == 0 && y[0] == 1 && y'[0] == 0,
   y[x], {x, -3, 10}];
Ai2 = NDSolve[y''[x] + x y[x] == 0 && y[0] == 0 && y'[0] == 1,
   y[x], {x, -3, 10}];
Ai = Join[Ai1, Ai2]
```

```
{{y[x] → InterpolatingFunction[{{-3., 10.}}, <>][x]},
 {y[x] → InterpolatingFunction[{{-3., 10.}}, <>][x]}}
```

```
Plot[Evaluate[y[x] /. Ai], {x, -3, 10}]
```



Euler

Euler

```
Rasterize[Plot3D[
  If[γ < 1 / 4, x^((1 - Sqrt[1 - 4 γ])/2), Sqrt[x] Cos[Log[x] Sqrt[4 γ - 1]]],
  {γ, 0, 4}, {x, 0, 1000},
  PlotPoints → 100, AxesLabel → Automatic, LabelStyle → Medium
 ], ImageSize → 500]
```

Euler




```
TagProperties["Euler"] =
  {ExportFormat → ".jpg", ExportOptions → {ImageSize → 800}};
CellExport["Euler"]
```



EulerDensity

```
Rasterize[DensityPlot[
  If[γ < 1 / 4, x^((1 - Sqrt[1 - 4 γ])/2),
   Sqrt[x] Cos[Log[x] Sqrt[4 γ - 1]]],
  {γ, 0, 4}, {x, 0, 1000},
  PlotPoints → 100, LabelStyle → Medium
 ], ImageSize → 360]
```

EulerDensity




```
TagProperties["EulerDensity"] =
  {ExportFormat → ".jpg", ExportOptions → {ImageSize → 800}};
CellExport["EulerDensity"]
```

Rasterize[DensityPlot[
  If[γ < 1 / 4, x^((1 - Sqrt[1 - 4 γ])/2),
   Sqrt[x] Cos[Log[x] Sqrt[4 γ - 1]]],
  {γ, 0, 4}, {x, 0, 1000},
  PlotPoints → 100, LabelStyle → Medium
 ], ImageSize → 360]



UltraFine

UltraFine

```
eq = x^2 y''[x] + γ y[x] /. y → (Y[e^#] &) /.
    x → Log[z];
eq = Expand[eq / Coefficient[eq, Y''[z]]]
```

UltraFine

$$\frac{\gamma\, Y[z]}{z^2\, \text{Log}[z]^2} + \frac{Y'[z]}{z} + Y''[z]$$

UltraFine

```
{p, q} = Coefficient[eq, #] & /@
    {Y'[z], Y[z]};
Q = q - 1/4 p^2 - 1/2 ∂_z p
```

UltraFine

$$\frac{1}{4\, z^2} + \frac{\gamma}{z^2\, \text{Log}[z]^2}$$

```
TagProperties["UltraFine"] = {ExportFormat → ".png"};
CellExport["UltraFine"]
```


```
eq = x^2 y''[x] + γ y[x] /. y → (Y[e^#] &) /.
    x → Log[z];
eq = Expand[eq / Coefficient[eq, Y''[z]]]
γ Y[z] / (z^2 Log[z]^2) + Y'[z] / z + Y''[z]
{p, q} = Coefficient[eq, #] & /@
    {Y'[z], Y[z]};
Q = q - 1/4 p^2 - 1/2 ∂_z p
1/(4 z^2) + γ / (z^2 Log[z]^2)
```

Bessel3D

Bessel3D

```
Rasterize[Plot3D[
  {BesselJ[α, x], BesselY[α, x]},
  {x, 0, 20}, {α, 0, 1},
  PlotPoints → 100, AxesLabel → Automatic, LabelStyle → Medium,
  PlotStyle → {Pink, LightBlue}
 ], ImageSize → 750]
```

Bessel3D



```
TagProperties["Bessel3D"] =
  {ExportFormat → ".jpg", ExportOptions → {ImageSize → 1200}};
CellExport["Bessel3D"]
```



Export All

```
False && (
  tags = Cases[
     NotebookGet[EvaluationNotebook[]],
     (CellTags → tag_) :→ tag,
     Infinity
    ] // Union;
  (# → CellExport[#]) & /@ tags
 )
```

```
False
```